<?xml version="1.0" encoding="UTF-8"?>

<!-- 
 * Copyright 2008 Philippe Prados.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 *
-->

<?xml-stylesheet type="text/xsl" href="../../../../xslt/xslt-to-xhtml.xslt" ?>

<!-- http://java.sun.com/products/jndi/serviceproviders.html#12 -->
<xsl:stylesheet
	version="2.0"
	xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"
	xsi:schemaLocation=
		"http://www.w3.org/1999/XSL/Transform http://www.w3.org/2007/schema-for-xslt20.xsd
		 http://jndi-resources.googlecode.com/1.0/ http://www.prados.fr/xsd/1.0/jndi-resources.xsd
		"
	xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
	xmlns:sch="http://www.w3.org/2001/XMLSchema"
	xmlns:jndi="http://jndi-resources.googlecode.com/1.0/"
	xmlns:tools="http://jndi-resources.googlecode.com/1.0/java/com.googlecode.jndiresources.tools.XSLTools"
	exclude-result-prefixes="#all"
>
<xsl:include href="../../../templates.xslt"/>
<xsl:include href="../../install-drivers.xslt"/>


<xsl:variable name="lib"><xsl:value-of select="concat($targetdir,'jboss.server.conf/lib/')"/></xsl:variable>

<xsl:template match="text()|comment()" />

<xsl:template match="jndi:resources[@id=$currentid]">
<xsl:value-of select="tools:info(concat('Generate ',$filename))"/>
<xsl:result-document  href="{$filename}">
<server>

<xsl:text>&#xA;</xsl:text>
<xsl:comment>Generated by jndi-resources v<xsl:value-of select="$version"/></xsl:comment>
<xsl:text>&#xA;</xsl:text>
<xsl:processing-instruction name="jndi-stylesheet">href="xslt/save.xslt" target="<xsl:value-of select="concat($basefilename,'.xml')"/>"</xsl:processing-instruction>
<xsl:text>&#xA;</xsl:text>
<xsl:apply-templates />

</server>
</xsl:result-document>
</xsl:template>

<xsl:template match="jndi:resources[@id=$currentid]/jndi:resource[@familly=$familly]">
	<xsl:value-of select="tools:fileCopy('../../../tools/xslt/save.xslt',concat($targetdir,'../xslt/'))" />

	<xsl:text>&#xA;   </xsl:text>
	<xsl:text>&#xA;   </xsl:text>
	<xsl:comment>Register external JNDI for key <xsl:value-of select="@name"/></xsl:comment>
	<xsl:text>&#xA;   </xsl:text>

	<mbean code="org.jboss.naming.ExternalContext" 
	       name="{$currentid}:service=ExternalContext,jndiName={@name}">
	    <attribute name="JndiName"><xsl:if test="$use-java-context='true'">java:</xsl:if><xsl:value-of select="concat('${jndi-prefix}',@name)"/></attribute>
	    <attribute name="InitialContext"><xsl:value-of select="$InitialContext"/></attribute>
	    <attribute name="RemoteAccess">true</attribute>
	    <attribute name="Properties">
<xsl:value-of select="$Properties"/>
<xsl:for-each select="jndi:extends[@appsrv=$appsrv]/*">
<xsl:value-of select="@name"/>=<xsl:value-of select="."/>
<xsl:text>&#xA;</xsl:text>
</xsl:for-each>
	    </attribute>
	    <depends>jboss:service=Naming</depends>
	</mbean>

	<xsl:call-template name="footer"/>
</xsl:template>

</xsl:stylesheet>
